


# TOSHIBA

Veá lo que puede hacer con un MSX



Basic-Parto

**TOSHIBA**




Impreso en el Japón



**Vea lo fácilmente  
que puede manejarlo  
"es muy divertido."  
Y cualquiera puede  
usarlo!!"**

HX-10



## Prefacio

Este librito muestra algunos sencillos programas para los usuarios del ordenador doméstico TOSHIBA MSX. Estos programas son suficientemente sencillos como para no requerir conocimientos de programación en BASIC.

Para una mejor comprensión de las órdenes e instrucciones, repase el Manual de Referencia del Basic que acompaña al ordenador. Observe como los programas cambian cuando escoja diferentes valores para las variables.

Introduzcamos estos programas y veamos que sucede:

- 1. Gráfico tridimensional**
- 2. Fantasía gráfica (1)**
- 3. Fantasía gráfica (2)**
- 4. Toque música con su MSX.**
- 5. Convierta su MSX en un instrumento musical.**
- 6. Convierta su MSX en un reloj.**

# 1 Gráfico Tridimensional



A pesar de su sencillez, este programa dibuja un cono nítido.


```

100 REM CONO
110 COLOR 15,0,1:SCREEN 2:CLS
120 M=RND(-TIME)
130 REM PRINCIPAL
140 C=RND(1)*15+1:IFC=16 THEN C=1
150 X=87:Y=120:R=70
160 GOSUB 500
170 C=RND(1)*15+1:IFC=16 THEN C=1
180 X=199:Y=56:R=50
190 GOSUB 500
200 GOTO 130
210 END
500 REM DIBUJO DE LOS CIRCULOS
510 FOR I=2 TO R STEP 2
520 RR=ABS(R-I):BEEP
530 CIRCLE(X,Y-I),RR,C
540 NEXT I
550 RETURN
  
```

Inicializaciones

Programa principal


Programa de dibujo



Este programa consta de las siguientes instrucciones:  
 Inicializaciones Líneas 100—120  
 Datos de los círculos Líneas 140—200  
 Programa de dibujo Líneas 500—550

Para cambiar el diámetro de las circunferencias altere el valor R en las líneas 150 y 180.  
 Para cambiar la distancia entre las circunferencias, altere el valor de STEP en la línea 510.  
 Para eliminar el sonido, borre la instrucción BEEP en la línea 520.  
 Ahora, a dibujar conos en la pantalla.

# 2 Dibuje una fantasía de cuadrados



Este programa creará un gráfico de cuadrados desplazados angularmente y con colores aleatorios.

```


100 REM CUADRADO
110 COLOR 15,0,1:SCREEN 2:CLS:DIM X(105),Y(105):N=4
120 XO=127:YO=95:LX=100:LY=95
130 X=RND(-TIME)
140 REM PRINCIPAL
150 M=(RND(1)*13 MOD 4)+1:IF N=M THEN 150
160 C=(RND(1)*46 MOD 15)+1
170 GOSUB 500
180 N=M:C=C+8:IF C>15 THEN C=C-15
190 COLOR 15,0,C
200 GOTO 140
210 END
500 REM SUBROUTINA PARA DIBUJAR LINEAS
510 IF M=1 THEN XX=XO:DX=-LX:YY=YO:DY=-LY
520 IF M=2 THEN XX=XO+1:DX=LX:YY=YO:DY=-LY
530 IF M=3 THEN XX=XO+1:DX=LX:YY=YO+1:DY=LY
540 IF M=4 THEN XX=XO:DX=-LX:YY=YO+1:DY=LY
550 X(1)=XX:Y(1)=YY
560 X(2)=XX+DX:Y(2)=YY
570 X(3)=XX+DX:Y(3)=YY+DY
580 X(4)=XX:Y(4)=YY+DY
590 LINE (X(1),Y(1))-(X(3),Y(3)),C,B
600 FOR T=1 TO 100
610 BEEP
620 J=T+1:X(T+4)=X(T):Y(T+4)=Y(T)
630 X(J)=X(J)+(X(J+1)-X(J))/10
640 Y(J)=Y(J)+(Y(J+1)-Y(J))/10
650 LINE(X(T),Y(T))-(X(T+1),Y(T+1)),C
660 NEXT
670 RETURN
  
```

Inicializaciones

Programa principal

Programa para dibujar líneas

Para eliminar el sonido, borre la instrucción BEEP en la línea 610. Admire los fantásticos cuadrados diversamente coloreados.



# 3 Dibuje gráficos de fantasía a



```

10 REM LINEA
20 SCREEN 2:COLOR 15,0,1:CLS
30 XO=127:YO=95:HI=200/192:L=85
40 DIM A(4),D(4)
100 REM PUESTA A CERO
110 X=RND(-TIME):CLS
120 FOR I=1 TO 4
130 A(I)=RND(1)*L/I+1
140 NEXT
200 REM CAMBIO DE MODELOS
210 C=(RND(1)*14 MOD 14)+2
220 FOR I=1 TO 4
230 D(I)=RND(1)*11-5
240 NEXT
300 REM PRINCIPAL
310 FOR I=1 TO 4
320 A(I)=A(I)+D(I)
330 A(I)=A(I)-INT(A(I)/L)*L
340 NEXT
350 GOSUB 500
360 IFRND(1)*5<1THEN200
370 IFRND(1)*40>1THEN300
380 FORI=0TO2000:NEXT
390 GOTO 100
400 END
500 REM SUBROUTINA PARA DIBUJAR LINEAS
510 FOR J=1 TO 4
520 IF J=1 THEN N=1:M=1
530 IF J=2 THEN N=-1:M=1
540 IF J=3 THEN N=-1:M=-1
550 IF J=4 THEN N=1:M=-1
560 X=A(1)*HI*N+XO:XD=A(2)*N*HI
570 Y=A(3)*M+YO :YD=A(4)*M
580 LINE(X,Y)-STEP(XD,YD),C
590 X=A(3)*HI*M+XO:XD=A(4)*HI*M
600 Y=A(1)*N+YO :YD=A(2)*N
610 LINE(X,Y)-STEP(XD,YD),C
620 NEXT
630 RETURN

```





Inicializaciones

Programa de puesta a cero

Programa de cambio de modelo

Programa principal

Subrutina para dibujar líneas



# 4 Programe "When the Saints Go Marching In"



La pauta para esta canción se muestra más abajo.

El programa de la canción es el siguiente:



```
10 PLAY"L4R4CEFG1"
20 PLAY"L4R4CEFG1"
30 PLAY"L4R4CEFL2GECED1"
40 PLAY"L4R4EDCC2.EEG2.F2.R4"
50 PLAY"L4R4EF6GE2DDE2CC2.R4"
60 GOTO 10
```

"Cambie la duración con L y el silencio con R"

```
PLAY "L2O3CDEFGABO4C" Pulse RETURN
PLAY "L8O3CDEFGABO4C" Pulse RETURN
```

En la pauta aparecen notas de varias longitudes (blanca, negra, corchea, etc.). A continuación se muestra como se ejecuta el HX-10.

Teclee "L" seguido de un entero en la gama 1 a 64 para especificar la duración de la nota.

(L1)	(L2)	(L4)	(L8)	(L16)	(L32)	(L64)
Redonda	Blanca	Negra	Corchea	Semicorchea	Fusa	Semifusa

```
PLAY "L8O3CDE4FGABO4C" Pulse RETURN
(Corchea, solamente (E) que está delante del número es una negra).
```

La combinación de estos dos métodos específicos permite una programación eficiente. Para programar música, use estas dos especificaciones.

Si en la canción que está programando existen varias notas de la misma duración, no hay necesidad de programarlas individualmente. Simplemente entre un número siguiendo a la letra L.

Puede especificar otras notas con solo números.

Para una nota punteada, entre un punto "." inmediatamente después de la especificación de la duración. Por ejemplo, si introduce "C4" el HX-10 tocará una nota C corchea punteada. El símbolo "#" representa una nota sostenida y "-" significa bemol.

Entrando "R" seguido por un número variará la duración de los silencios.

PLAY "CDEFr4GAB" Pulse RETURN

(R1)	(R2)	(R4)	(R8)	(R16)	(R32)	(R64)

Pongamos más notas a esta melodía.

```
10 REM MUSICA
20 CLS:SCREEN 2
30 LINE(80,30)-(160,90),1,BF
40 OPEN"GRP:" FOR OUTPUT AS#1
50 PSET(85,60),1:COLOR 14
60 PRINT #1,"M U S I C"
100 RESTORE 200
110 FOR I=1 TO 5
120 READ A#,B#
125 PLAY A#,B#
130 NEXT
140 GOTO 100
150 END
200 REM
210 DATA "t170s13m5014","t170s13m5011"
220 DATA "df+gar1df+gar4r2","r4r2ddd"
230 DATA "df+ga2f+2d2f+2er4r2","ddda"
240 DATA "f+f+ed2.df+2aaag2.,""addg"
250 DATA "r4bba2f+2d2e2d1","gdd2a2d"
```

<b>Signo</b>	<b>Margen</b>
O < Número > (Tono)	1—8
L < Número > (Duración)	1—64
R < Número > (Silencio)	1—64
T < Número > (Tempo)	32—255
V < Número > (Volumen)	0—15

**Otros signos**  
 + (Más): #  
 - (Menos): b  
 · (Punto): Nota punteada

Ahora, usted es un compositor por ordenador. Intente otras canciones.

# 5 Convierta su HX-10 en un instrumento musical



Este programa convertirá el teclado en un teclado de piano.

```

10 REM PIANO
20 SCREEN 2:COLOR 15,4,4:CLS
30 K$="AWSDRFTGHUJIKOL;['"+CHR$(30)+CHR$(31)
40 K$=K$+"awsdrftghujikol"
50 O=0:GOSUB 500
60 REM PRINCIPAL
70 BEEP:LINE(108,25)-STEP(20,15),14,BF
80 IF O>3 THEN O=3
90 IF O<-1 THEN O=-1
100 PSET(108,30),14:PRINT#1,USING"+#";O
110 PLAY"T200L16S0M15000"
200 A$=INKEY$:IF A$="" THEN 200
210 S=INSTR(K$,A$):IF S=0 THEN 200
220 IF S>20 THEN S=S-20
230 IF S=19 THEN O=O+1:GOTO 60
240 IF S=20 THEN O=O-1:GOTO 60
250 X=S+O*12+32:PLAY"N=X;"
260 GOTO 200
270 END
500 REM SUBROUTINA PARA EL DIBUJO DE LAS TECLAS
510 OPEN "GRP:" FOR OUTPUT AS #1
520 LINE(34,60)-(209,130),1,BF
530 X$="ASDFGHJKL;['":COLOR 1
540 FOR I=0 TO 10
550 X=35+I*16
560 LINE(X,61)-(X+14,129),14,BF
570 PSET(X+5,110),14:PRINT#1,MID$(X$,I+1,1)
580 NEXT
590 X$="W RT UID ['":COLOR 15
600 FOR I=0 TO 9
610 IF I=1 OR I=4 OR I=8 THEN 650
620 X=43+I*16
630 LINE(X,60)-(X+14,95),1,BF
640 PSET(X+5,80),1:PRINT#1,MID$(X$,I+1,1)
650 NEXT
660 LINE(34,25)-(150,40),14,BF:COLOR 1
670 PSET(40,30):PRINT#1,"OCTAVE [ ]"
680 RETURN

```

Inicializaciones


Programa principal

Programa para la visualización de las teclas

Este programa le permite cambiar de octava desde -2 hasta +3 por medio de las teclas de flecha. Si usted desea modificar el tempo, cambie el valor T200 por T50 en la línea 110 del programa. Esto cambia la longitud de las notas. Para cambiar el color de fondo del teclado, altere los valores de la línea 20.

**Ejemplo: Color 15, 4, 4—Color 15, 8, 8**

Ahora, ha convertido el MSX en un instrumento musical. ¿Porque no compone alguna pieza y la ejecuta?



# 6 Convierta su MSX en un reloj




Esto es un reloj analógico.

Pulse la tecla F1 para la puesta en hora. Luego use las teclas F2 y F3 para poner la hora y las teclas F4 y F5 para los minutos. Finalmente pulse la tecla RETURN.

En la pantalla aparecerá automáticamente un reloj.

Cambie el color de fondo para experimentar. Por ejemplo, cambiando color 15, 4, 1 a color 15, 8, 1 en la línea 20 y B=4 a B=8 en la línea 30 aparecerá un reloj rojo.



```

10 REM RELOJ
20 COLOR 15,4,1:SCREEN 2:CLS:C=15:B=4
30 C=15:B=4:X0=127:Y0=80
40 R0=70:R1=40:R2=55:R3=R1-10:R4=R1/2
50 PI=3.1415926535898#
60 H=10:M=10:XM=0:YM=0:XH=0:YH=0
70 GOSUB 1000
80 ONKEYGOSUB 500,700,730,750,770
100 REM PRINCIPAL
110 PUT SPRITE 1,(X0-3,Y0-R1-3),8:BEEP
120 S=0:ON INTERVAL=50 GOSUB 200
130 GOSUB 300
140 INTERVAL ON:KEY(1) ON
150 GOTO 150
160 END
200 REM SUBROUTINA VISUALIZACION SEGUNDOS
210 S=S+1:RS=PI*1.5+S*PI/30
220 X=COS(RS)*R1+X0-3:Y=SIN(RS)*R1+Y0-3
230 PUT SPRITE 1,(X,Y),8:BEEP
240 IF S<60 THEN RETURN
250 S=0:M=M+1:IF M<60 THEN 280
260 M=0:H=H+1:IF H<12 THEN 280
270 H=0
280 GOSUB 300
290 RETURN
300 REM SUBROUTINA VISUALIZACION HORAS Y MINUTOS
310 LINE(X0,Y0)-STEP(XM,YM),B
320 RM=PI*1.5+M*PI/30
330 XM=COS(RM)*R3:YM=SIN(RM)*R3
340 LINE(X0,Y0)-STEP(XM,YM),C
350 LINE(X0,Y0)-STEP(XH,YH),B
360 RH=PI*1.5+H*PI/6+M*PI/360
370 XH=COS(RH)*R4:YH=SIN(RH)*R4
380 LINE(X0,Y0)-STEP(XH,YH),C
390 CIRCLE(X0,Y0),3,C:PAINT(X0+1,Y0)
400 RETURN
500 REM SUBROUTINA DE PUESTA EN HORA
510 INTERVAL OFF:KEY(1)OFF

```

Inicializaciones

Programa principal

Subrutina para  
visualizar los segundos

Subrutina para  
visualizar las horas  
y los minutos

```

520 LINE(20,170)-(247,190),B,BF
530 PSET(25,172),B
540 PRINT#1,"[F2] [F3] [F4] [F5] [RETURN]"
550 PSET(25,182),B
560 PRINT#1," Hr+ Hr- Mi+ Mi- PUESTA H"
570 FOR L=2 TO 5:KEY(L)ON:NEXT
580 LINE(40,150)-(120,159),B,BF
590 PSET(50,151),B:PRINT#1,USING"###:##";H;M
600 K#=INKEY#:IF K#="" THEN 600
610 IF K#<>CHR$(13) THEN 600
620 LINE(40,150)-(120,159),B,BF
630 FOR L=2 TO 5:KEY(L)OFF:NEXT
640 GOSUB 2000
650 RETURN 100
700 REM
710 H=H+1:IF H>11 THEN H=11
720 RETURN 580
730 H=H-1:IF H<0 THEN H=0
740 RETURN 580
750 M=M+1:IF M>59 THEN M=59
760 RETURN 580
770 M=M-1:IF M<0 THEN M=0
780 RETURN 580
1000 REM SUBROUTINA PARA EL DIBUJO INICIAL
1010 SPRITE$(1)=CHR$(16)+"8:8"+CHR$(16)
1020 OPEN"grp:" FOR OUTPUT AS#1
1030 X1=X0-R0-20:Y1=Y0-R0-10
1040 X2=X0+R0+20:Y2=Y0+R0+10
1050 LINE(X1,Y1)-(X2,Y2),0,B
1060 PAINT(10,10),0
1070 SPRITE$(1)=CHR$(16)+"8:8"+CHR$(16)
1080 CIRCLE(X0,Y0),R0,C
1090 FOR I=-90 TO 270 STEP 5
1100 D=PI*I/180
1110 X=COS(D)*R1+X0:Y=SIN(D)*R1+Y0
1120 PSET(X,Y),C
1130 IF (I MOD 30)<>0 THEN 1150
1140 CIRCLE(X,Y),2,C
1150 NEXT I
1160 FOR I=1 TO 12
1170 D=PI*(I-3)/6
1180 X=COS(D)*R2+X0-10:Y=SIN(D)*R2+Y0-3
1190 PRESET(X,Y),B
1200 PRINT#1,I
1210 NEXT I
1220 GOSUB 2000
1230 RETURN
2000 REM
2010 LINE(20,170)-(247,190),0,BF:PSET(38,180),0
2020 PRINT#1,"[F1] --- PUESTA EN HORA"
2030 RETURN

```

Subrutina para poner  
en hora el reloj.

Programa de cuenta  
del tiempo.

Subrutina para  
visualizar la imagen  
inicial.

Ahora usted ya conoce algunas de las muchas cosas interesantes que puede hacer con su TOSHIBA HX-10. ¿Cual fué el programa que más le gustó? Familiarícese con el BASIC. Le abrirá las puertas del excitante mundo de los ordenadores. El Ordenador Doméstico TOSHIBA \*MSX, un amigo con el que puede contar.

\*MSX es una marca registrada por Microsoft Corporation.

